Integration 1

▶ The `ratint` function

▶ Various rational functions

Exercise 1

When x is large, y is approximately $\frac{x^4}{4}$. Indeed, the formula for y is as follows:

```
> y := ratint(((x^3+1)/(x^2+1))^3,x);
```

$$y := \frac{x^4}{4} - \frac{3x^2}{2} + 3x + \frac{15x^3}{4(x^2+1)^2} + \frac{x^2}{2(x^2+1)^2} + \frac{13x}{4(x^2+1)^2} + \frac{1}{(x^2+1)^2} + 3\ln(x^2+1) - \frac{21\arctan(x)}{4} \quad \textbf{(1)}$$

The first term is $\frac{x^4}{4}$, and the other terms are much smaller, at least when x is large. Indeed, when x is large, all of the terms with x^2+1 on the bottom will be very small. Also, you should recall from the theory of trigonometric functions that $|\arctan(x)|$ is always at most $\frac{\mathrm{Pi}}{2}$, so this will also be negligible

in comparison to $\frac{x^4}{4}$ for large x. The terms $3x$ and $\frac{3x^2}{2}$ will get large as x does, but much more slowly than $\frac{x^4}{4}$ does. For example, when $x=1000$, the term $\frac{x^4}{4}$ is of the order of a trillion, whereas $\frac{3x^2}{2}$ is of the order of a million, and $3x$ is only 3000. The behaviour of $\ln(x^2+1)$ is a little less obvious,
but in fact it grows much more slowly even than x, as you could check by plotting the graph.

We can ask Maple to separate the terms as follows:

```
> terms := [op(y)];
```

$$terms := \left[\frac{x^4}{4}, -\frac{3x^2}{2}, 3x, \frac{15x^3}{4(x^2+1)^2}, \frac{x^2}{2(x^2+1)^2}, \frac{13x}{4(x^2+1)^2}, \frac{1}{(x^2+1)^2}, 3\ln(x^2+1), -\frac{21\arctan(x)}{4}\right] \quad \textbf{(2)}$$

We can then check the relative size when $x=1000$:

```
> evalf(subs(x=1000,terms));
```

$[2.500000000\ 10^{11}, -1.500000000\ 10^{6}, 3000., 0.003749992500, 4.999990000\ 10^{-7},$ **(3)**

$3.249993500\ 10^{-9}, 9.999980000\ 10^{-13}, 41.44653468, -8.241430717]$

You can see that the first term is much bigger than the rest. We can also see this by plotting.
In the picture below, the red curve is $\frac{x^4}{4}$, and the other terms are plotted in various different colours.
By $x = 5$, the red curve is already much bigger than the others. If you change the range to **1..100** (say) then only the red curve is visible because all other terms are squashed down onto the x axis.

```
> plot(terms,x=1..5);
```



We can also just plot y and $\frac{x^4}{4}$ together. If we stop at $x = 9$ then you can still see the difference between
the two curves, but if we went much further than that then they would be indistinguishable.

```
> plot([y,x^4/4],x=-9..9);
```




```
> unassign('y','terms');
```

Exercise 2

Here is the formula for y:

```
> y := ratint((a*x^2+b)/(c*x^2+d),x);
```

$$y := \frac{a\,x}{c} - \frac{\arctan\left(\frac{c\,x}{\sqrt{d\,c}}\right)a\,d}{c\sqrt{d\,c}} + \frac{\arctan\left(\frac{c\,x}{\sqrt{d\,c}}\right)b}{\sqrt{d\,c}} \quad \textbf{(4)}$$

This is $y = \frac{a\,x}{c}$ (which gives a line of slope $\frac{a}{c}$) plus some other terms, which will be much smaller when x is large. One way to see this is to look at $\frac{dy}{dx}$ when x is large, or in other words, the limit of $\frac{dy}{dx}$ as x tends to infinity. Of course $\frac{dy}{dx}$ is just the function we first thought of, namely $\frac{a\,x^2+b}{c\,x^2+d}$. When x is very large, b and d will be negligible compared with $a\,x^2$ and $c\,x^2$, so the function is approximately $\frac{a\,x^2}{c\,x^2} = \frac{a}{c}$.

```
> limit(diff(y,x),x=infinity);
```

$$\frac{a}{c} \quad \textbf{(5)}$$

We can see this graphically if we choose some numbers for a, b, c and d.

```
> plot(subs(a=7,b=-3,c=2,d=8,[y,a*x/c]),x=10..100);
```




```
> unassign('y');
```

Exercise 3

(a) It is true that the integral of a rational function can contain terms like $a \ln(x^2 + u\,x + v)$. For example:

```
> y := (8*x+8)/(x^2+2*x+3);
```

$$y := \frac{8\,x+8}{x^2+2\,x+3} \tag{6}$$

```
> ratint(y,x);
```

$$4\ln(x^2+2\,x+3) \tag{7}$$

or more generally:

```
> y := a*(2*x+u)/(x^2+u*x+v);
```

$$y := \frac{a\,(2\,x+u)}{u\,x+x^2+v} \tag{8}$$

```
> ratint(y,x);
```

$$a\ln(u\,x+x^2+v) \tag{9}$$

(b) It is *not* true that terms like $x\ln(x+u)$ can occur in the integral of a rational function $g(x)$. If there were a term $x\ln(x+u)$ in $\int g(x)\,\mathrm{d}x$, then there would have to be a term $\frac{\partial}{\partial x}\,(x\ln(x+u))$ in $\frac{\mathrm{d}}{\mathrm{d}x}\left(\int g(x)\,\mathrm{d}x\right)$. In other words, there would be a term

$\ln(x+u) + \dfrac{x}{x+u}$ in $g(x)$, which is not allowed, as $g(x)$ is supposed to be a rational function.

(c) For essentially the same reason, there cannot be any terms like $a \ln(x+u)^2$ in the integral of a rational function, because $\dfrac{\partial}{\partial x}\left(a \ln(x+u)^2\right) = \dfrac{2\, a \ln(x+u)}{x-u}$, which is not allowed as a term in a rational function.

Exercise 4

If $b^2 - 4\,c < 0$ then the function $x^2 + b\,x + c$ has no real roots, and the integral $\displaystyle\int \frac{1}{x^2 + b\,x + c}\, dx$ involves
the arctan( ) function. Here is an example:

```
> b := 2; c := 3; b^2-4*c;
```

$$b := 2$$

$$c := 3$$

$$-8 \tag{10}$$

```
> plot(x^2+b*x+c,x=-2..1,0..6);
```




```
> ratint(1/(x^2+b*x+c),x);
```

$$\frac{\sqrt{2}\,\arctan\left(\frac{\sqrt{2}\,x}{2} + \frac{\sqrt{2}}{2}\right)}{2} \tag{11}$$

If $b^2 - 4\,c = 0$ then the function $x^2 + b\,x + c$ has only one real root (where the graph touches the x axis but does not cross it) and the integral involves neither arctan( ) nor ln( ); in fact, it just
has the form $-\dfrac{1}{x-u}$. Here is an example:

```
> b := 4; c := 4; b^2-4*c;
```

$$b := 4$$

$$c := 4$$

$$0 \tag{12}$$

```
> plot(x^2+b*x+c,x=-4..1,0..6);
```




```
> ratint(1/(x^2+b*x+c),x);
```

$$-\frac{1}{x+2} \tag{13}$$

If $0 < b^2 - 4\,c$ then the function $x^2 + b\,x + c$ has two real roots, and the integral is a sum of two terms involving ln( ). Here is an example:

```
> b := -5; c := 4; b^2-4*c;
```

$$b := -5$$

$$c := 4$$

$$9 \tag{14}$$

```
> plot(x^2+b*x+c,x=-1..5);
```




```
> ratint(1/(x^2+b*x+c),x);
```

$$-\frac{\ln(|x-1|)}{3}+\frac{\ln(|x-4|)}{3} \quad \textbf{(15)}$$

```
> unassign('b','c');
```

Exercise 5

Terms like $\ln(|x-u|)$ in $\int g(x)\,\mathrm{d}x$ are associated with places where $g(x)$ blows up to infinity (which are known as the *poles* of $g(x)$). More precisely, if there is a term $\ln(|x-u|)$ in the integral, then $g(x)$ *must* blow up at $x=u$. Conversely, if $g(x)$ blows up at $x=u$ then there will *usually* be a term $\ln(|x-u|)$ in the integral, but not always; there may instead be terms of the form $(x-u)^{-n}$. The functions $g_3(x)$ and $g_7(x)$ are examples of this.

Here we consider $g_2(x)$; there are poles at $x=-1$ and $x=1$, and corresponding terms $\ln(|x+1|)$ and $\ln(|x-1|)$ in the integral.

```
> y := g[2](x); z := ratint(y,x);
```

$$y := \frac{1}{x-1}+\frac{1}{x^2-1}+\frac{1}{x^3-1}$$

$$z := \frac{11\ln(|x-1|)}{6}-\frac{\ln(|x+1|)}{2}-\frac{\ln\left(x^2+x+1\right)}{6}-\frac{\sqrt{3}\arctan\left(\frac{2\sqrt{3}\,x}{3}+\frac{\sqrt{3}}{3}\right)}{3} \quad \textbf{(16)}$$

```
> plot([y,z],x=-5..5,-10..10);
```



Here we consider $g_3(x)$. There are no $\ln(\ )$ terms in the integral, but nonetheless there are poles at $x=1$, $x=2$ and $x=3$.

```
> y := g[3](x); z := ratint(y,x);
```

$$y := \frac{1}{(x-1)^2} + \frac{1}{(x-2)^3} + \frac{1}{(x-3)^4}$$

$$z := -\frac{1}{3\,(x-3)^3} - \frac{1}{2\,(x-2)^2} - \frac{1}{x-1} \tag{17}$$

```
> plot([y,z],x=0..5,-100..100,numpoints=400);
```



Here we consider $g_4(x)$. There is a pole at $x = -1$, and a corresponding term $\ln(|x+1|)$ in the integral.

```
> y := g[4](x); z := ratint(y,x);
```

$$y := \frac{1}{x+1} + \frac{1}{x^2+1} + \frac{1}{x^3+1}$$

$$z := \frac{4\ln(|x+1|)}{3} + \arctan(x) - \frac{\ln(x^2 - x + 1)}{6} + \frac{\sqrt{3}\arctan\left(\frac{2\sqrt{3}\,x}{3} - \frac{\sqrt{3}}{3}\right)}{3} \qquad \textbf{(18)}$$

```
> plot([y,z],x=-5..5,-10..10);
```



Exercise 6

We only get terms like $\frac{1}{x-u}$ or $\frac{1}{(x-u)^n}$ in $\int g(x)\,\mathrm{d}x$ if the denominator of $g(x)$ has repeated roots.

A repeated root at $x=u$ gives a factor $(x-u)^n$ in the denominator, with $1<n$. This gives a term $\frac{1}{(x-u)^{n-1}}$ in the integral (and possibly also some terms like $\frac{1}{(x-u)^k}$ for smaller values of k).

Here we consider $g_3(x)$:

```
> y := g[3](x);
```

$$y := \frac{1}{(x-1)^2} + \frac{1}{(x-2)^3} + \frac{1}{(x-3)^4} \tag{19}$$

The denominator has repeated roots at $x=1$, $x=2$ and $x=3$:

```
> d := denom(factor(y));
```

$$d := (x-1)^2\,(x-2)^3\,(x-3)^4 \tag{20}$$

These show up in the graph of d as points where the graph touches the x axis without crossing it, or where it crosses but the tangent line at the crossing point is horizontal:

```
> plot(d,x=0..4,-0.3...0.3);
```



The integral has terms of the form $\frac{1}{(x-u)^n}$ for $u = 1, 2$ and 3.

```
> ratint(y,x);
```

$$-\frac{1}{3\,(x-3)^3}-\frac{1}{2\,(x-2)^2}-\frac{1}{x-1} \tag{21}$$

Here we consider $g_4(x)$:

```
> y := g[4](x);
```

$$y := \frac{1}{x+1}+\frac{1}{x^2+1}+\frac{1}{x^3+1} \tag{22}$$

The denominator has a root at $x = -1$, which is not repeated:

```
> d := denom(factor(y));
```

$$d := \left(x^2+1\right)\,(x+1)\,\left(x^2-x+1\right) \tag{23}$$

This is visible in the graph of d. Note that where the graph meets the x axis, the tangent line is not horizontal, corresponding to the fact that the root is not repeated.

```
> plot(d,x=-2..2,-10..10);
```



The integral has no terms of the form $\frac{1}{(x-u)^n}$.

```
> ratint(y,x);
```

$$\frac{4\ln(|x+1|)}{3}+\arctan(x)-\frac{\ln(x^2-x+1)}{6}+\frac{\sqrt{3}\arctan\left(\frac{2\sqrt{3}\,x}{3}-\frac{\sqrt{3}}{3}\right)}{3} \quad (24)$$

Here we consider $g_6(x)$:

```
> y := g[6](x);
```

$$y := \frac{x\,(x+2)\,(x+4)}{(x+1)^2\,(x+3)^2} \quad (25)$$

The denominator has repeated roots at $x=-3$ and $x=-1$:

```
> d := denom(factor(y));
```

$$d := (x+1)^2\,(x+3)^2 \quad (26)$$

These visible in the graph of d, as points where the curve meets the x axis without crossing it. Note that the tangent line is horizontal at these points.

```
> plot(d,x=-4..2,-10..10);
```



The integral has terms of the form $\frac{1}{x-u}$ for $x = -3$ and $x = -1$.

```
> ratint(y,x);
```

$$-\frac{3}{4\,(x+3)} + \frac{\ln(|x+3|)}{2} + \frac{3}{4\,(x+1)} + \frac{\ln(|x+1|)}{2} \tag{27}$$

We now integrate some random rational functions, and observe that we do not get any terms like $\frac{1}{(x-u)^n}$. The reason is simply that the denominator is a random polynomial, which will almost never have repeated roots. The roots of a random polynomial are randomly distributed, so it would be a big coincidence for two roots to be in the same place, and this does not often happen.

```
> r := randpoly(x)/randpoly(x);
```

$$r := \frac{-7\,x^5 + 22\,x^4 - 55\,x^3 - 94\,x^2 + 87\,x - 56}{-62\,x^4 + 97\,x^3 - 73\,x^2 - 4\,x - 83} \tag{28}$$

```
> ratint(r,x);
```

$$-0.1781997919\,x - 0.1572050640\,\ln(x^2 + 0.6287014875 + 0.6984629175\,x) + 0.9550059466\,\arctan(1.404778618\,x + 0.4905928858) + 0.3948880051\,\ln(x^2 + 2.129324814 - 2.262979046\,x) + 0.8455509582\,\arctan(1.085254918\,x - 1.227954570) + 0.05645161290\,x^2 \tag{29}$$

```
> r := randpoly(x)/randpoly(x);
```

$$r := \frac{-10\,x^5 + 62\,x^4 - 82\,x^3 + 80\,x^2 - 44\,x + 71}{-17\,x^5 - 75\,x^4 - 10\,x^3 - 7\,x^2 - 40\,x + 42} \quad (30)$$

```
> ratint(r,x);
```

$$0.5882352941\,x + 0.1061874592\,\ln\left(x^2 + 0.8275212923 - 0.3675458439\,x\right) + 0.5261842533\,\arctan(1.122427911\,x - 0.2062718568) - 0.4428128431\,\ln(|x - 0.6164177376|) - 7.992432621\,\ln(|x + 4.258350792|) + 1.980656010\,\ln(|x + 1.137377494|) \quad (31)$$

```
> r := randpoly(x)/randpoly(x);
```

$$r := \frac{-50\,x^5 + 23\,x^4 + 75\,x^3 - 92\,x^2 + 6\,x + 74}{72\,x^5 + 37\,x^4 - 23\,x^3 + 87\,x^2 + 44\,x + 29} \quad (32)$$

```
> ratint(r,x);
```

$$-0.6944444444\,x - 0.009138551763\,\ln\left(x^2 + 1.143500883 - 1.314095787\,x\right) - 0.03032363307\,\arctan(1.185289343\,x - 0.7787918661) + 0.3440274514\,\ln\left(x^2 + 0.2672062237 + 0.5097812758\,x\right) + 1.542925025\,\arctan(2.223666777\,x + 0.5667918431) + 0.006533929131\,\ln(|x + 1.318203400|) \quad (33)$$